

THE TRANSVERSE SPACE-CHARGE FORCE IN TRI-GAUSSIAN DISTRIBUTION

K.Y. Ng

Fermi National Accelerator Laboratory, P.O. Box 500, Batavia, IL 60510*

(October 21, 2005)

(Revised October 15, 2007)

Abstract

The transverse space-charge force exerted on a particle inside a beam with transverse bi-Gaussian distribution is formulated for all situations, for a flat beam, for an almost round beam, for the particle near the beam center, and far away from the beam center. A subroutine to compute the space-charge force is given.

*Operated by the Fermi Research Alliance, under contract with the U.S. Department of Energy.

1 INTRODUCTION

In tracking, the transverse space-charge force can be represented by changes in the horizontal and vertical divergences, $\Delta x'$ and $\Delta y'$ at many locations around the accelerator ring. In this note, we are going to list some formulas for $\Delta x'$ and $\Delta y'$ arising from space-charge kicks when the beam is tri-Gaussian distributed. We will discuss separately a flat beam and an almost round beam.

We are not interested in the situation when the emittance growth arising from space charge becomes too large and the shape of the beam becomes weird. For this reason, we can assume the bunch still retains its tri-Gaussian distribution, with its rms sizes σ_x , σ_y , and σ_z increasing by certain factors. Thus after each turn, σ_x , σ_y , and σ_z can be re-calculated. The electric potential for a particle of charge e at location x, y, z is therefore given by the formula, [1]

$$U_{sc}(x, y, z) = \frac{eN_b}{4\pi^{3/2}\epsilon_0} \int_0^\infty dt \frac{\exp\left[-\frac{x^2}{2\sigma_x^2+t} - \frac{y^2}{2\sigma_y^2+t} - \frac{z^2}{2\sigma_z^2+t}\right] - 1}{\sqrt{(2\sigma_x^2+t)(2\sigma_y^2+t)(2\sigma_z^2+t)}}, \quad (1.1)$$

where N_b is the number of particles in the bunch. The force acting on the particle is just $-e\vec{\nabla}U_{sc}$. Written in this way, the potential vanishes at the center of the bunch. To include the magnetic part of the force, we just need to add γ^2 in the denominator, where γ is the relativistic factor. There is no closed form for this integral.

2 Linear Approximation

When incorporating this into a code, it will be nice if there is an option of including only the linear part of the space-charge force. Take the x -component of the force. After performing the derivative with respect to x , the linear part can be obtained by setting $x = y = 0$ in the integrand. Thus

$$F_{x \text{ linear}} = -e \frac{\partial U_{sc}}{\partial x} \Big|_{\text{linear}} = \frac{e^2 N_b x}{2\pi^{3/2} \epsilon_0 \gamma^2} \int_0^\infty dt \frac{\exp\left[-\frac{z^2}{2\sigma_z^2+t}\right]}{(2\sigma_x^2+t)^{3/2} (2\sigma_y^2+t)^{1/2} (2\sigma_z^2+t)^{1/2}}. \quad (2.1)$$

We can be sure that the algebraic sign on the right side is correct, because the space-charge force is repulsive and therefore positive in the positive x -direction. So far as we know, this

integral cannot be performed in the closed form. However, since σ_y and σ_x are both very much smaller than σ_z in a particle accelerator ring, an approximation can be made. Notice that the integration variable t varies mostly between 0 and $2\sigma_x^2$. We can therefore make the replacement

$$2\sigma_z^2 + t \longrightarrow 2\sigma_z^2. \quad (2.2)$$

The above reduces to

$$F_{x \text{ linear}} = \frac{e^2 N_b x}{2\pi\epsilon_0\gamma^2} \frac{e^{-z^2/(2\sigma_z^2)}}{\sqrt{2\pi}\sigma_z} \int_0^\infty dt \frac{1}{(2\sigma_x^2 + t)^{3/2}(2\sigma_y^2 + t)^{1/2}}. \quad (2.3)$$

This integral can now be performed in the closed form to give the familiar result

$$F_{x \text{ linear}} = -e \frac{\partial U_{sc}}{\partial x} \Big|_{\text{linear}} = \frac{e^2 x}{2\pi\epsilon_0\gamma^2} \frac{1}{\sigma_x(\sigma_x + \sigma_y)} \frac{N_b e^{-z^2/(2\sigma_z^2)}}{\sqrt{2\pi}\sigma_z}. \quad (2.4)$$

In the same way, the linear part of the vertical space-charge force on the particle is

$$F_{y \text{ linear}} = -e \frac{\partial U_{sc}}{\partial y} \Big|_{\text{linear}} = \frac{e^2 y}{2\pi\epsilon_0\gamma^2} \frac{1}{\sigma_y(\sigma_x + \sigma_y)} \frac{N_b e^{-z^2/(2\sigma_z^2)}}{\sqrt{2\pi}\sigma_z}. \quad (2.5)$$

The equations of motion in the transverse planes are

$$\begin{aligned} x'' + K_x x &= \frac{F_x}{\gamma m v^2} = -\frac{e}{\gamma m v^2} \frac{\partial U_{sc}}{\partial x}, \\ y'' + K_y y &= \frac{F_y}{\gamma m v^2} = -\frac{e}{\gamma m v^2} \frac{\partial U_{sc}}{\partial y}, \end{aligned} \quad (2.6)$$

where $K_x x$ and $K_y y$ are restoring forces from the magnetic elements, while $F_{x,y}$ are the horizontal and vertical forces arising from space-charge effects. In above, m is the particle mass and $v = \beta c$ is the nominal velocity of the beam particles. Integrating over a length L of the orbit where the beam radii do not change much, we obtain the changes in x' and y' coming from space-charge effects only,

$$\begin{aligned} \Delta x' &= -\frac{eL}{\gamma m v^2} \frac{\partial U_{sc}}{\partial x}, \\ \Delta y' &= -\frac{eL}{\gamma m v^2} \frac{\partial U_{sc}}{\partial y}. \end{aligned} \quad (2.7)$$

Thus, the linearized space-charge force leads to

$$\begin{aligned} \Delta x' &= \frac{2N_b r_0 L e^{-z^2/(2\sigma_z^2)}}{\gamma^3 \beta^2 \sqrt{2\pi}\sigma_z} \frac{x}{\sigma_x(\sigma_x + \sigma_y)}, \\ \Delta y' &= \frac{2N_b r_0 L e^{-z^2/(2\sigma_z^2)}}{\gamma^3 \beta^2 \sqrt{2\pi}\sigma_z} \frac{y}{\sigma_y(\sigma_x + \sigma_y)}, \end{aligned} \quad (2.8)$$

where $r_0 = e^2/(4\pi\epsilon_0 mc^2)$ represents the classical radius of the beam particle. They can also be rewritten as

$$\begin{aligned}\Delta x' &= K_{sc}L \frac{x}{\sigma_x(\sigma_x + \sigma_y)}, \\ \Delta y' &= K_{sc}L \frac{y}{\sigma_y(\sigma_x + \sigma_y)}.\end{aligned}\quad (2.9)$$

where

$$K_{sc} = \frac{2N_b r_0 e^{-z^2/(2\sigma_z^2)}}{\gamma^3 \beta^2 \sqrt{2\pi} \sigma_z}, \quad (2.10)$$

is called the *local space-charge perveance* of the beam, with $N_b e^{-z^2/(2\sigma_z^2)}/\sqrt{2\pi}\sigma_z$ being the local linear beam density.

3 Application to Electron Bunches

An electron bunch usually has its length very much larger than the transverse radii. Thus the replacement in Eq. (2.2) can be made, and the space-charge potential becomes (including electric and magnetic contributions)

$$U_{sc}(x, y, z) = \frac{eN_b}{4\pi^{3/2}\epsilon_0\gamma^2} \frac{e^{-z^2/(2\sigma_z^2)}}{\sqrt{2}\sigma_z} \int_0^\infty dt \frac{\exp\left[-\frac{x^2}{2\sigma_x^2+t} - \frac{y^2}{2\sigma_y^2+t}\right] - 1}{\sqrt{(2\sigma_x^2+t)(2\sigma_y^2+t)}}. \quad (3.1)$$

This integral can be performed in the closed form in terms of the complex error function. [2] As will be shown below, however, the analytic expression is useful only for electron bunches, where the horizontal beam radius is very much larger than the vertical ($\sigma_x \gg \sigma_y$).

Introduce the following new variables:

$$s^2 = \frac{2\sigma_y^2 + t}{2\sigma_x^2 + t}, \quad a = \frac{x}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}}, \quad b = \frac{y}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}}, \quad \text{and} \quad r = \frac{\sigma_y}{\sigma_x}. \quad (3.2)$$

It is then easy to obtain

$$U_{sc}(x, y, z) = \frac{e^2 N_b e^{-z^2/(2\sigma_z^2)}}{(2\pi)^{3/2} \epsilon_0 \gamma^2 \sigma_z} \int_r^1 \frac{ds}{1-s^2} \left[e^{-a^2(1-s^2) - b^2(\frac{1}{s^2} - 1)} - 1 \right]. \quad (3.3)$$

We therefore have

$$\begin{aligned}\Delta x' &= K_{sc}L \frac{x}{\sigma_x^2 - \sigma_y^2} \int_r^1 ds e^{-a^2(1-s^2) - b^2(\frac{1}{s^2} - 1)}, \\ \Delta y' &= K_{sc}L \frac{y}{\sigma_x^2 - \sigma_y^2} \int_r^1 \frac{ds}{s^2} e^{-a^2(1-s^2) - b^2(\frac{1}{s^2} - 1)}.\end{aligned}\quad (3.4)$$

We understand that the signs before the right sides of these two equations are correct, because for positive x and/or y , the horizontal and/or vertical divergence should increase.

The changes in the horizontal and vertical divergences can be combined as a complex variable

$$\Delta x' - i\Delta y' = \frac{K_{sc}L}{\sqrt{\sigma_x^2 - \sigma_y^2}} \int_r^1 ds \left(a - i\frac{b}{s^2} \right) e^{-(a+ib)^2 + (as+ib/s)^2}.\quad (3.5)$$

A new variable of integration

$$\zeta = as + i\frac{b}{s}\quad (3.6)$$

is now introduced, which simplifies the above to

$$\Delta x' - i\Delta y' = K_{sc}L \frac{e^{-(a+ib)^2}}{\sqrt{\sigma_x^2 - \sigma_y^2}} \int_{ar + \frac{ib}{r}}^{a+ib} e^{\zeta^2} d\zeta.\quad (3.7)$$

From the definition of the complex error function

$$w(z) = e^{-z^2} \left[1 + \frac{2i}{\sqrt{\pi}} \int_0^z e^{\zeta^2} d\zeta \right],\quad (3.8)$$

we arrive at

$$\Delta x' - i\Delta y' = -i \frac{\sqrt{\pi} K_{sc}L}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}} \left[w(a+ib) - e^{-a^2(1-r^2) - b^2(\frac{1}{r^2} - 1)} w\left(ar + i\frac{b}{r}\right) \right],\quad (3.9)$$

or

$$\begin{aligned}\Delta x' &= \frac{\sqrt{\pi} K_{sc}L}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}} \mathcal{I}m \left[w\left(\frac{x+iy}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}}\right) - e^{-\frac{x^2}{2\sigma_x^2} - \frac{y^2}{2\sigma_y^2}} w\left(\frac{x\frac{\sigma_y}{\sigma_x} + iy\frac{\sigma_x}{\sigma_y}}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}}\right) \right], \\ \Delta y' &= \frac{\sqrt{\pi} K_{sc}L}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}} \mathcal{R}e \left[w\left(\frac{x+iy}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}}\right) - e^{-\frac{x^2}{2\sigma_x^2} - \frac{y^2}{2\sigma_y^2}} w\left(\frac{x\frac{\sigma_y}{\sigma_x} + iy\frac{\sigma_x}{\sigma_y}}{\sqrt{2(\sigma_x^2 - \sigma_y^2)}}\right) \right].\end{aligned}\quad (3.10)$$

These expressions appear to diverge when $\sigma_x = \sigma_y$ because of the factors outside the squared-brackets. In fact, this is not true, because the expressions inside the square brackets will

provide zeros at $\sigma_x = \sigma_y$ to cancel the poles outside. This is obvious, because the original expression for the space-charge potential in Eq. (1.1) is well-behaved at $\sigma_x = \sigma_y$.

In case $\sigma_x < \sigma_y$, the above can be easily transformed to

$$\Delta y' - i\Delta x' = -i \frac{\sqrt{\pi} K_{sc} L}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}} \left[w(a' + ib') - e^{-a'^2(1-r'^2) - b'^2\left(\frac{1}{r'^2} - 1\right)} w\left(a'r' + i\frac{b'}{r'}\right) \right], \quad (3.11)$$

with

$$a' = \frac{y}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}}, \quad b' = \frac{x}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}}, \quad \text{and} \quad r' = \frac{\sigma_x}{\sigma_y}. \quad (3.12)$$

Thus

$$\begin{aligned} \Delta x' &= \frac{\sqrt{\pi} K_{sc} L}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}} \operatorname{Re} \left[w\left(\frac{y + ix}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}}\right) - e^{-\frac{x^2}{2\sigma_x^2} - \frac{y^2}{2\sigma_y^2}} w\left(\frac{y\frac{\sigma_x}{\sigma_y} + ix\frac{\sigma_y}{\sigma_x}}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}}\right) \right], \\ \Delta y' &= \frac{\sqrt{\pi} K_{sc} L}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}} \operatorname{Im} \left[w\left(\frac{y + ix}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}}\right) - e^{-\frac{x^2}{2\sigma_x^2} - \frac{y^2}{2\sigma_y^2}} w\left(\frac{y\frac{\sigma_x}{\sigma_y} + ix\frac{\sigma_y}{\sigma_x}}{\sqrt{2(\sigma_y^2 - \sigma_x^2)}}\right) \right]. \end{aligned} \quad (3.13)$$

4 ALMOST ROUND BEAM

It is obvious that the expressions in terms of the complex error function in Eq. (3.10) cannot be applied when the beam is almost round because of the singularities at $\sigma_x = \sigma_y$ outside and within the square brackets. Instead, let us start our discussion from Eq. (3.4), which can be rewritten as

$$\begin{aligned} \Delta x' &= K_{sc} L \frac{x}{\sigma_x^2} f_x(x^2, y^2), \\ \Delta y' &= K_{sc} L \frac{y}{\sigma_y^2} f_y(x^2, y^2), \end{aligned} \quad (4.1)$$

where

$$\begin{aligned} f_x(x^2, y^2) &= \frac{1}{1 - r^2} \int_r^1 ds e^{-a^2(1-s^2) - b^2\left(\frac{1}{s^2} - 1\right)}, \\ f_y(x^2, y^2) &= \frac{r^2}{1 - r^2} \int_r^1 \frac{ds}{s^2} e^{-a^2(1-s^2) - b^2\left(\frac{1}{s^2} - 1\right)}, \end{aligned} \quad (4.2)$$

with $r = \sigma_y/\sigma_x$. We may perform power expansion in x^2 and y^2 to obtain

$$\begin{aligned}
f_x(x^2, y^2) &= \frac{1}{1+r} \left[1 - \left(\frac{x^2}{2\sigma_x^2} \right) \frac{2+r}{3(1+r)} - \left(\frac{y^2}{2\sigma_y^2} \right) \frac{r}{1+r} + \left(\frac{x^2}{2\sigma_x^2} \right)^2 \frac{8+9r+3r^2}{30(1+r)^2} + \right. \\
&\quad \left. + \left(\frac{y^2}{2\sigma_y^2} \right)^2 \frac{r(1+3r)}{6(1+r)^2} + \left(\frac{x^2}{2\sigma_x^2} \right) \left(\frac{y^2}{2\sigma_y^2} \right) \frac{r(3+r)}{3(1+r)^2} + \dots \right], \\
f_y(x^2, y^2) &= \frac{r}{1+r} \left[1 - \left(\frac{x^2}{2\sigma_x^2} \right) \frac{1}{1+r} - \left(\frac{y^2}{2\sigma_y^2} \right) \frac{1+2r}{3(1+r)} + \left(\frac{x^2}{2\sigma_x^2} \right)^2 \frac{3+r}{6(1+r)^2} + \right. \\
&\quad \left. + \left(\frac{y^2}{2\sigma_y^2} \right)^2 \frac{3+9r+8r^2}{30(1+r)^2} + \left(\frac{x^2}{2\sigma_x^2} \right) \left(\frac{y^2}{2\sigma_y^2} \right) \frac{1+3r}{3(1+r)^2} + \dots \right]. \quad (4.3)
\end{aligned}$$

It is evident that $f_y(x^2, y^2)$ can be obtained from $f_x(x^2, y^2)$ by suitable replacements of r by $1/r$. Unfortunately, these formulas may not be very useful, because they lead to the incorrect consequence that the space-charge force increases without limit when the particle is far away from the beam center. Instead we should expand Eq. (4.2) around a round beam. In other words, we should perform Taylor's expansion in the variable ϵ , where

$$\epsilon = 1 - r = \frac{\sigma_x - \sigma_y}{\sigma_x}. \quad (4.4)$$

Let $u = 1 - s$, then the x -component of Eq. (4.2) becomes

$$f_x(x^2, y^2) = \frac{1}{(1+r)\epsilon} \int_0^\epsilon du \exp \left[-\frac{x^2 u(1-u/2)}{\sigma_x(\sigma_x + \sigma_y)\epsilon} - \frac{y^2 u(1-u/2)}{\sigma_x(\sigma_x + \sigma_y)\epsilon(1-u)^2} \right]. \quad (4.5)$$

We further introduce $t = u/\epsilon$, so that

$$f_x(x^2, y^2) = \frac{\sigma_x}{\sigma_x + \sigma_y} \int_0^1 dt \exp \left[-\frac{x^2 t(1-\epsilon t/2)}{\sigma_x(\sigma_x + \sigma_y)} - \frac{y^2 t(1-\epsilon t/2)}{\sigma_x(\sigma_x + \sigma_y)(1-\epsilon t)^2} \right]. \quad (4.6)$$

Now ϵ appears only in the integrand and a power series can be easily obtained. One may argue that ϵ also appears in σ_y implicitly. But we would like to leave σ_y as is; otherwise, the expansion will become more messy. It is easy to see that the above can be expressed in terms of the integrals

$$f_n(w) = \int_0^1 e^{-wt} t^n dt = \frac{n!}{w^{n+1}} \left(1 - e^{-w} \sum_{k=0}^n \frac{w^k}{k!} \right), \quad (4.7)$$

where

$$w = p + q, \quad p = \frac{x^2}{\sigma_x(\sigma_x + \sigma_y)}, \quad q = \frac{y^2}{\sigma_x(\sigma_x + \sigma_y)}. \quad (4.8)$$

In fact,

$$f_x(x^2, y^2) = \frac{\sigma_x}{\sigma_x + \sigma_y} \left\{ f_0(w) + \epsilon \frac{p-3q}{2} f_2(w) + \epsilon^2 \left[-2q f_3(w) + \frac{(p-3q)^2}{8} f_4(w) \right] + \epsilon^3 \left[-\frac{5q}{2} f_4(w) + \frac{3q^2 - 2pq}{2} f_5(w) + \frac{(p-3q)^3}{48} f_6(w) \right] + \mathcal{O}(\epsilon^4) \right\}. \quad (4.9)$$

In the same way, we obtain

$$f_y(x^2, y^2) = \frac{\sigma_y^2}{\sigma_x(\sigma_x + \sigma_y)} \left\{ f_0(w) + \epsilon \left[2f_1(w) + \frac{p-3q}{2} f_2(w) \right] + \epsilon^2 \left[3f_2(w) + (p-5q)f_3(w) + \frac{(p-3q)^2}{8} f_4(w) \right] + \epsilon^3 \left[4f_3(w) + \frac{3p-22q}{2} f_4(w) + \frac{p^2+15q^2-10pq}{2} f_5(w) + \frac{(p-3q)^3}{48} f_6(w) \right] + \mathcal{O}(\epsilon^4) \right\}. \quad (4.10)$$

The expression on the right side of Eq. (4.7) has an apparent singularity at $w = 0$, or when the particle resides at the center of the beam. This apparent singularity can be easily removed by the following expansion whenever w is small:

$$f_n(w) = \sum_{k=0}^{\infty} \frac{(-w)^k}{(n+k+1)k!}. \quad (4.11)$$

Note that the above expressions are valid independent of whether ϵ is positive or negative.

With the above formulas, the space-charge force has been coded in the Fortran subroutine `spchforce(x, y, sx, sy, Fx, Fy, icode)` and is listed in the Appendix, where one inputs (\mathbf{x}, \mathbf{y}) as the transverse position of the particle in a bi-Gaussian distribution with rms spreads $(\mathbf{sx}, \mathbf{sy}) = (\sigma_x, \sigma_y)$. The outputs `Fx` and `Fy` give the horizontal and vertical kicks on the particle:

$$\begin{aligned} \Delta x' &= K_{sc} L F_x, \\ \Delta y' &= K_{sc} L F_y, \end{aligned} \quad (4.12)$$

where K_{sc} is the local space-charge perveance of the beam while σ_z is its longitudinal rms spread. The output variable `icode` serves as a record of how the space-charge force is computed. While `icode`= 1 or 2 denotes the use of the complex-error-function expressions of Eq. (3.10) or (3.13), `icode`= 3 or 4 denotes the use of the small- ϵ expansion of Eqs. (4.9) and (4.10) through the exact integration result of Eq. (4.7) or through the small- w expansion of Eq. (4.11).

APPENDIX

In below, we list the subroutine `spchforce(x,y,sx,sy,Fx,Fy,icode)`. In compilation, `cernlib` is required for the complex error function. Note that in the subroutine the vertical coordinate is denoted by z instead of y .

```

subroutine spchforce(x,z,sx,sz,fx,fz,icode)
c   Space charge force in a bi-Gaussian beam
c   Input:  x,z are horizontal and vertical particle positions
c           sx,sz are horizontal and vertical rms spread of
c           transverse distribution, assumed to be bi-Gaussian
c   Output: Fx,Fz are horizontal and vertical components of sp-ch
c           force. When multiplied by Ksc*L, the beam local sp-ch
c           perveance give Delta x' and Delta z' for a length L.
c           Ksc=2*N*r0/(gamma^3*\beta^2), N=local beam linear
c           density, r0=classical particle radius.
c           index tracks how the force is computed. When complex error
c           fcn is used, index=1 or 2 implies sx>sz or sx<sz.
c           When expansion in ep is used, index=3 or 4 implies
c           exact integration or expansion around beam center.
implicit real*8(a-b,d-h,o-z)
implicit complex*16(c)
complex*16 wwerf
spi=1.772453851d0    !sqrt(pi)
r=sz/sx
ep=1.d0-r
if(abs(ep).gt.0.05d0)go to 1
c   Use expansion when ep is small
p=x*x/(sx*(sx+sz))
q=z*z/(sx*(sx+sz))
w=p+q
if(w.gt.0.01d0)then
  expw=exp(-w)
  w2=w*w
  w3=w*w2

```

```

w4=w2*w2
w5=w4*w
w6=w5*w
f0=(1.d0-expw)/w
f1=(1.d0-expw*(w+1.d0))/w2
f2=(2.d0-expw*(w2+2.d0*w+2.d0))/w3
f3=(6.d0-expw*(w3+3.d0*w2+6.d0*w+6.d0))/w4
f4=(24.d0-expw*(w4+4.d0*w3+12.d0*w2+24.d0*w+24.d0))/w5
f5=(120.d0-expw*(w5+5.d0*w4+20.d0*w3
*      +60.d0*w2+120.d0*w+120.d0))/w6
f6=(720.d0-expw*(w6+6.d0*w5+30.d0*w4
*      +120.d0*w3+360.d0*w2+720.d0*w+720.d0))/(w6*w)
icode=3
else
f0=1.d0      -w/2.d0+w2/6.d0  -w3/24.d0+w4/120.d0
f1=0.5d0     -w/3.d0+w2/8.d0  -w3/30.d0+w4/144.d0
f2=1.d0/3.d0-w/4.d0-w2/10.d0-w3/36.d0+w4/168.d0
f3=1.d0/4.d0-w/5.d0+w2/12.d0-w3/42.d0+w4/192.d0
f4=1.d0/5.d0-w/6.d0+w2/14.d0-w3/48.d0+w4/216.d0
f5=1.d0/6.d0-w/7.d0+w2/16.d0-w3/54.d0+w4/240.d0
f6=1.d0/7.d0-w/8.d0+w2/18.d0-w3/60.d0+w4/264.d0
icode=4
endif
fx=f0+ep*f2*(p-3.d0*q)/2.d0
*      +ep*ep*(f4*(p-3.d0*q)**2/8.d0-2.d0*f3*q)
*      +ep*ep*ep*(f6*(p-3.d0*q)**3/48.d0+f5*(3.d0*q*q-2.d0*p*q)/2.d0
*      -2.5d0*f4*q)
fz=f0+ep*(2.d0*f1+(p-3.d0*q)*f2/2.d0)
*      +ep*ep*(3.d0*f2+(p-5.d0*q)*f3+(p-3.d0*q)**2*f4/8.d0)
*      +ep*ep*ep*(4.d0*f3+(1.5d0*p-11.d0*q)*f4
*      +(0.25d0*p*p+3.75d0*q*q-2.5d0*p*q)*f5
*      +(p-3.d0*q)**3*f6/48.d0)
fx=fx*x/(sx*(sx+sz))
fz=fz*z/(sx*(sx+sz))
return

```

```

c      Use complex error function
1      continue
      if(ep.gt.0.d0)then
          diff=sqrt(2.d0*(sx*sx-sz*sz))
          w1=x/diff
          w2=z/diff
      else
          diff=sqrt(2.d0*(sz*sz-sx*sx))
          w2=x/diff
          w1=z/diff
          r=1.d0/r
      endif
      ca=cplx(w1,w2)
      cb=cplx(w1*r,w2/r)
      ckick=wwerf(ca)-wwerf(cb)*exp(-x*x/(2.d0*sx*sx)-z*z/(2.d0*sz*sz))
      if(ep.gt.0.d0)then
          fx=imag(ckick)*spi/diff
          fz=real(ckick)*spi/diff
          icode=1
          return
      else
          fz=imag(ckick)*spi/diff
          fx=real(ckick)*spi/diff
          icode=2
          return
      endif
      end
end

```

References

- [1] A derivation is available at K. Takayama, Lett. Al Nuovo Cimento **34**, 190 (1982).
- [2] M. Bassetti and G.A. Erskine, *Closed Expression for the Electrical Field of a Two-Dimensional Gaussian Charge*, CERN Report CERN-ISR-TH/80-06, 1980.